# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Leewings Ng <osideal@gmail.com>
# Contributor: Hiroshi Yui <hiroshiyui@gmail.com>

pkgname=ibus-table-chinese
pkgver=1.8.12
pkgrel=1
pkgdesc="Some table-based input method of tables engines for IBus"
arch=('any')
url="https://github.com/mike-fabian/ibus-table-chinese"
license=('GPL3')
depends=('ibus-table')
makedepends=('cmake-fedora')
source=("$pkgname-$pkgver.tar.gz::https://github.com/mike-fabian/ibus-table-chinese/archive/$pkgver.tar.gz")
sha512sums=('418a93709e6e7cd56cb0251fedcce8c3640ccce54d822ff5051491c8b5b1ee7d547cea018019b91c1a091aa5243c347c4c9fdb66992b8dabf12ceb6719300a85')

build() {
  cd $pkgname-$pkgver
  cmake . -DCMAKE_FEDORA_TMP_DIR=. -DDATA_DIR=/usr/share -DPRJ_DOC_DIR=/usr/share/doc/$pkgname -DCMAKE_INSTALL_PREFIX=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
